

Excel für Microsoft 365

Relative und absolute Bezüge



Inhaltsverzeichnis

Einleitung.....	2
Relativer oder absoluter Bezug?	2
Mischbezüge	6
Schnittmengen berechnen	8
Die Z1S1-Bezugsart.....	8

Abbildungsverzeichnis

Abb. 1: <i>Die Ausgangstabelle für die Arbeitszeitberechnung</i>	2
Abb. 2: <i>Die erste (noch fehlerhafte) Zeitberechnung für die Arbeitszeit</i>	3
Abb. 3: <i>Das korrekte Ergebnis der Zeitberechnung</i>	4
Abb. 4: <i>Die Tabelle mit berechnetem Arbeitslohn</i>	5
Abb. 5: <i>Die Ausgangstabelle für die fortlaufende Summierung</i>	5
Abb. 6: <i>Die Ergebnistabelle mit der fortlaufenden Summierung</i>	6
Abb. 7: <i>Die Ausgangstabelle für das kleine Einmaleins</i>	7
Abb. 8: <i>Die komplette Tabelle mit dem kleinen Einmaleins</i>	7
Abb. 9: <i>Die Schnittmenge zweier Zellbereiche</i>	8
Abb. 10: <i>Dialogfeld Excel-Optionen, Kategorie Formeln</i>	9
Abb. 11: <i>Eine Formel in der Z1S1-Bezugsart</i>	9

Einleitung

Wenn Sie Formeln in Ihren Datenlisten, Tabellen, usw. einsetzen und sich dabei auf den Inhalt von Tabellenzellen beziehen, werden Sie üblicherweise die Zellennamen über die Kombination aus Spaltenbuchstaben und Zeilennummer (also z.B. **F3**) angeben. Dabei handelt es sich um die sogenannte **relative Bezugsart**. In den meisten Fällen, wo Sie mit Formeln in Excel arbeiten, werden Sie auch problemlos mit dieser Bezugsart arbeiten können. Auch dann, wenn Sie die Tabellenzellen mit den Formeln verschieben, kopieren oder mit dem *automatischen Ausfüllen* auf andere Tabellenzellen übertragen. Es gibt aber auch Fälle, da müssen Sie in einer Formel sich konkret auf eine ganz bestimmte Tabellenzelle beziehen. In diesem Fall müssen Sie die **absolute Bezugsart** verwenden. In diesem Skript wird der Unterschied zwischen relativen und absoluten Bezügen gezeigt und welche Besonderheiten es diesbezüglich noch gibt. Das Skript bezieht sich auf [Excel für Microsoft 365](#), kann aber auch bei älteren Excel-Versionen (auf alle Fälle ab Version 2016) verwendet werden.

Relativer oder absoluter Bezug?

Wenn Sie bisher in Excel immer nur mit der relativen Bezugsart gearbeitet haben, werden Sie sich jetzt vielleicht fragen, wann soll ich denn die absolute Bezugsart einsetzen? Und wie teile ich Excel mit, wenn ich einen absoluten Bezug in einer Formel einsetzen will? Die Antworten auf diese Fragen sollen konkret an einem Beispiel gezeigt werden. In einer Tabelle (siehe Abbildung 1) sind mehrere Mitarbeiterinnen und Mitarbeiter (Spalte **A**) eingetragen und die Uhrzeit für den Arbeitsbeginn (Spalte **B**) und das Arbeitsende (Spalte **C**). Nun soll für jede Person die reine Arbeitszeit (Spalte **D**) ermittelt werden. Dafür müssen Sie den *Arbeitsbeginn* vom *Arbeitsende* subtrahieren. Außerdem soll eine Zeitangabe für die *Mittagspause* (Tabellenzelle **E1**) vom Ergebnis abgezogen werden. Wenn Excel die reine Arbeitszeit ermittelt hat, soll anschließend auch noch der Arbeitslohn berechnet werden.

	A	B	C	D	E	F	G	H	I
1	Stundenlohn:	14,50 €		Mittagspause:	45 Min.		Inhalt von E1: 00:45:00		
2							Zahlenformat von E1: mm "Min."		
3	Mitarbeiter*in	Arbeitsbeginn	Arbeitsende	reine Arbeitszeit	Arbeitslohn				
4	Schmidt	07:45	15:55						
5	Becker	07:41	15:40						
6	Mayer	08:01	16:10						
7	Volkman	08:05	16:05						
8	Müller	07:37	15:32						

Abb. 1: Die Ausgangstabelle für die Arbeitszeitberechnung

Als erstes wählen Sie im vorliegenden Beispiel die Tabellenzelle **D4** aus und tragen die Formel

=C4-B4-E1

ein und bestätigen die Eingabe. Sie erhalten das Ergebnis **07:25**. Dabei handelt es sich nicht um eine Uhrzeit, sondern um die Anzahl der Stunden und Minuten, die die Person *Schmidt* gearbeitet hat. Jetzt wollen Sie für die anderen Personen die Formel kopieren und verwenden dafür am besten das Verfahren *Automatisches Ausfüllen* (siehe Skript [Excel für Microsoft 365 – Automatisches Ausfüllen](#)). Sie wählen also erneut die Tabellenzelle **D4** aus, bewegen das Mausymbol auf das kleine grüne Ausfüllpünktchen in der rechten unteren Ecke des Auswahlrahmens und führen einen Doppelklick mit der linken Maustaste aus. Das Ergebnis sehen Sie in Abbildung 2, Seite 3.

	A	B	C	D	E	F	G	H	I
1	Stundenlohn:	14,50 €		Mittagspause:	45 Min.		Inhalt von E1: 00:45:00		
2							Zahlenformat von E1: mm "Min."		
3	<u>Mitarbeiter*in</u>	<u>Arbeitsbeginn</u>	<u>Arbeitsende</u>	<u>reine Arbeitszeit</u>	<u>Arbeitslohn</u>				
4	Schmidt	07:45	15:55	07:25					
5	Becker	07:41	15:40	07:59					
6	Mayer	08:01	16:10	#WERT!					
7	Volkmann	08:05	16:05	08:00					
8	Müller	07:37	15:32	07:55					

Abb. 2: Die erste (noch fehlerhafte) Zeitberechnung für die Arbeitszeit

Was Ihnen sicherlich sofort an der Tabelle auffällt, ist der Fehlerwert **#WERT!** in Tabellenzelle **D6**. Sie werden sich sicherlich fragen, wodurch ist dieser Fehlerwert entstanden? Was habe ich falsch gemacht? Bevor diese Frage beantwortet wird, noch kurz der Hinweis, dass Sie in den Tabellenzellen **D5**, **D7** und **D8** Zeitangaben sehen, diese aber nicht korrekt sind. Insbesondere beim Mitarbeiter *Volkmann* können Sie das sehr leicht auch so direkt erkennen. Als Ergebnis sehen Sie **08:00**. Wenn Sie sich den Arbeitsbeginn (**08:05**) und das Arbeitsende (**16:05**) ansehen, stellen Sie ganz leicht fest, dass die Differenz zwischen den beiden Zeitangaben genau **8** Stunden beträgt, was ja auch in Tabellenzelle **D7** steht. Mit anderen Worten: die Mittagspause wurde nicht von der Differenz der beiden Zeitangaben abgezogen. Aber warum nicht? Schauen wir uns für die Beantwortung der Fragen nochmals die Formel in Tabellenzelle **D4** an:

=C4-B4-E1 (Tabellenzelle **D4**)

Wie Sie ja bereits wissen, handelt es sich bei der Angabe der Zellennamen um *relative Bezüge*. Das bedeutet in diesem konkreten Fall, dass (ausgehend von Tabellenzelle **D4**, wo die Formel eingetragen ist) zunächst der erste Zellennamen (**C4**; *Arbeitsende*) relativ gesehen eine Tabellenzelle links neben der Tabellenzelle mit der Formel steht. Der nächste Zellennamen (**B4**; *Arbeitsbeginn*) steht zwei Tabellenzellen links der Formel. Und der dritte Zellennamen (**E1**; *Mittagspause*) steht drei Tabellenzellen oberhalb und eine Tabellenzelle rechts bezogen auf die Tabellenzelle mit der Formel. Beim Kopieren der Formel in die Tabellenzellen **D5** bis **D8** mit dem automatischen Ausfüllen sind die relativen Bezüge erhalten geblieben. Das bedeutet für die anderen Mitarbeiter, dass folgende Formeln in den entsprechenden Tabellenzellen stehen:

=C5-B5-E2 (Tabellenzelle **D5**)

=C6-B6-E3 (Tabellenzelle **D6**)

=C7-B7-E4 (Tabellenzelle **D7**)

=C8-B8-E5 (Tabellenzelle **D8**)

Bezogen auf die Tabellenzellen mit dem Arbeitsende und dem Arbeitsbeginn (**C5** bis **C8** und **B5** bis **B8**) ist alles in Ordnung. Dort ist die relative Bezugsart korrekt. Aber die Angabe der Mittagspause stimmt nicht. Für den Mitarbeiter *Becker* nimmt Excel den Inhalt der Tabellenzelle **E2**, die aber leer ist (entspricht dem Wert **0**). Also bereits bei diesem Mitarbeiter wurde die Mittagspause nicht von der Zeitdifferenz abgezogen. Beim Mitarbeiter *Mayer* hat Excel die Tabellenzelle **E3** genommen. In dieser Tabellenzelle steht aber Text (**Arbeitslohn**) und Text kann bei einer mathematischen Berechnung nicht verwendet werden, daher der Fehlerwert **#WERT!**. Auch bei den beiden Mitarbeitern *Volkmann* und *Müller* kommen falsche Ergebniswerte heraus (bei beiden fehlt wieder die Mittagspause). Für die Mittagspause muss also die **absolute Bezugsart** genommen werden. Denn für alle Mitarbeiter steht der Wert für die Mittagspause immer in der Tabellenzelle **E1**. Und das muss beim Kopieren der

Formel entsprechend berücksichtigt werden. Zunächst werden alle Formeln in den Tabellenzellen **D4** bis **D8** gelöscht und anschließend in die Tabellenzelle **D4** folgende Formel eingetragen:

=C4-B4-\$E\$1

In dem Sie bei der Angabe der Tabellenzelle **E1** vor den Spaltenbuchstaben und vor der Zeilennummer jeweils ein Dollarzeichen (\$) schreiben, ändern Sie die Bezugsart dieser Zellangabe von **relativ** auf **absolut**. Wenn Sie jetzt die Formel nach unten kopieren, stehen in den Tabellenzellen **D5** bis **D8** folgende Formeln:

=C5-B5-\$E\$1

=C6-B6-\$E\$1






=C7-B7-\$E\$1


=C8-B8-\$E\$1

Sie sehen also, für die Mittagspause wird immer die Tabellenzelle **E1** genommen. Damit wäre das Problem gelöst. Das korrekte Ergebnis sehen Sie in Abbildung 3.

	A	B	C	D	E	F	G	H	I
1	Stundenlohn:	14,50 €		Mittagspause:	45 Min.		Inhalt von E1: 00:45:00		
2							Zahlenformat von E1: mm "Min."		
3	Mitarbeiter*in	Arbeitsbeginn	Arbeitsende	reine Arbeitszeit	Arbeitslohn				
4	Schmidt	07:45	15:55	07:25					
5	Becker	07:41	15:40	07:14					
6	Mayer	08:01	16:10	07:24					
7	Volkmann	08:05	16:05	07:15					
8	Müller	07:37	15:32	07:10					

Abb. 3: Das korrekte Ergebnis der Zeitberechnung

Es gibt noch eine interessante Vorgehensweise, wie Sie schnell aus der relativen Bezugsart eine absolute Bezugsart machen können. Nehmen wir nochmal die Ausgangstabelle aus Abbildung 1, Seite 2. Jetzt geben Sie in die Tabellenzelle **D4** die Formel **=C4-B4-E1** ein und bestätigen die Eingabe, ohne darauf zu achten, dass die relative Zellangabe **E1** ja eigentlich falsch ist. Erst nachdem Sie die Formel in die darunterliegenden Tabellenzellen kopiert haben, stellen Sie Ihren Fehler fest. Sie wollen den Fehler korrigieren, in dem Sie die Tabellenzelle **D4** auswählen und in die Zellbearbeitung wechseln (z.B. mit der Funktionstaste ). Sie setzen die Schreibmarke beim Zellennamen **E1** vor den Buchstaben **E** und geben das Dollarzeichen ein. Dann setzen Sie die Schreibmarke vor die Zeilennummer **1** und geben erneut das Dollarzeichen ein. Diese Vorgehensweise ist nur bei einem einzigen Zellennamen vielleicht noch akzeptabel. Aber stellen Sie sich vor, in der Formel haben Sie gleich ein Dutzend Zellennamen, bei denen Sie die absolute Bezugsart benötigen. Jedes Mal die Schreibmarke an die entsprechende Stelle setzen und das Dollarzeichen eingeben ist auf Dauer zu mühselig. Es geht einfacher: setzen Sie die Schreibmarke auf den Zellennamen in der Formel und drücken die Funktionstaste . Sofort wechselt Excel von der relativen zur absoluten Bezugsart. Drücken Sie die Funktionstaste  erneut, verschwindet das Dollarzeichen vor dem Spaltenbuchstaben. Sie haben jetzt einen Mischbezug. Nochmal  drücken und das Dollarzeichen steht vor dem Spaltenbuchstaben, aber nicht vor der Zeilennummer (ebenfalls ein Mischbezug). Benutzen Sie nochmal , haben Sie wieder die relative Bezugsart (ohne Dollarzeichen).

Anmerkung: Wenn Sie der Tabellenzelle **E1** (aus dem vorherigen Beispiel) einen benutzerdefinierten Namen geben (z.B. **Mittagspause**; siehe Skript **Excel für Microsoft 365 – Zellen benennen**) und diesen Namen in der Formel verwenden (also **=C4-B4-Mittagspause**), handelt es sich immer um eine absolute Bezugsart und Sie können sich die Sache mit dem Dollarzeichen und der Funktionstaste  ersparen.

Als Ergänzung soll die Beispieltabelle noch dadurch erweitert werden, dass aus der reinen Arbeitszeit für jeden Mitarbeiter der Arbeitslohn berechnet wird. Der Stundenlohn steht in der Tabellenzelle **B1**. Bei der Berechnung des Arbeitslohns muss Excel aus der berechneten Arbeitszeit den Stunden- und den Minutenwert ermitteln und mit dem Stundenlohn entsprechenden verrechnen. Dafür stellt Excel die Funktionen **STUNDE** und **MINUTE** zur Verfügung. Die Formeln für die Tabellenzellen **E4** bis **E8** haben folgendes Aussehen (das Ergebnis sehen Sie in Abbildung 4; für die Ergebniswerte wurde das Zahlenformat **##0,00 €** genommen):

=STUNDE(D4)*\$B\$1+MINUTE(D4)*\$B\$1/60 (Tabellenzelle **E4**)
=STUNDE(D5)*\$B\$1+MINUTE(D5)*\$B\$1/60 (Tabellenzelle **E5**)
=STUNDE(D6)*\$B\$1+MINUTE(D6)*\$B\$1/60 (Tabellenzelle **E6**)
=STUNDE(D7)*\$B\$1+MINUTE(D7)*\$B\$1/60 (Tabellenzelle **E7**)
=STUNDE(D8)*\$B\$1+MINUTE(D8)*\$B\$1/60 (Tabellenzelle **E8**)

	A	B	C	D	E	F	G	H	I
1	Stundenlohn:	14,50 €		Mittagspause:	45 Min.		Inhalt von E1: 00:45:00		
2							Zahlenformat von E1: mm "Min."		
3	Mitarbeiter*in	Arbeitsbeginn	Arbeitsende	reine Arbeitszeit	Arbeitslohn				
4	Schmidt	07:45	15:55	07:25	107,54 €				
5	Becker	07:41	15:40	07:14	104,88 €				
6	Mayer	08:01	16:10	07:24	107,30 €				
7	Volkmann	08:05	16:05	07:15	105,13 €				
8	Müller	07:37	15:32	07:10	103,92 €				

Abb. 4: Die Tabelle mit berechnetem Arbeitslohn

Bei einem weiteren Beispiel soll eine fortlaufende Summierung von Daten vorgenommen werden. Als Beispiel sollen die Summen von verkauften Objekten über einen bestimmten Zeitraum ermittelt werden. In Abbildung 5 sehen Sie fünf Spalten (**B** bis **F**) für die Objekte und die verkauften Stückzahlen über den Zeitraum von *Januar* bis *Juni*.

	A	B	C	D	E	F	G
1		Objekt 1	Objekt 2	Objekt 3	Objekt 4	Objekt 5	fortlaufende Summe
2	Januar	335	333	350	326	357	
3	Februar	282	365	396	332	238	
4	März	343	219	214	380	308	
5	April	294	218	304	229	373	
6	Mai	309	281	333	361	277	
7	Juni	377	282	210	237	285	

Abb. 5: Die Ausgangstabelle für die fortlaufende Summierung

In der Spalte **G** sollen nun die fortlaufenden Summenwerte der verkauften Objekte ermittelt werden. Für den Monat *Januar* sollen nur die Werte im Bereich **B2:F2** berechnet werden. Für den Monat *Februar* wird der Bereich **B2:F3** für die Summe genommen. Beim Monat *März* ist es der Bereich **B2:F4**, usw. Wenn Sie für die erste Summe in der Tabellenzelle die Formel **=SUMME(B2:F2)** eingeben und bestätigen, erhalten Sie das korrekte Ergebnis. Wenn Sie jetzt allerdings die Formel mit dem *automatischen Ausfüllen* nach unten in die Tabellenzelle **G3** bis **G7** kopieren, bekommen Sie zwar auch Ergebniswerte, aber immer nur die Summen der Werte für jeden einzelnen Monat. Sie erhalten also keine fortlaufende Summierung. Um die fortlaufende Summierung zu erhalten, tragen Sie in die Tabellenzelle **G2** folgende Formel ein und bestätigen die Eingabe:

=SUMME(\$B\$2:F2)

Wenn Sie anschließend die Formel mit dem automatischen Ausfüllen nach unten kopieren, erhalten Sie das gewünschte Ergebnis, da nur die zweite Zellangabe neu angepasst wird (also aus **F2** wird dann **F3**, **F4**, usw.). Die erste Zellangabe (**\$B\$2**) dagegen bleibt unverändert, da es sich ja um einen absoluten Bezug handelt. Abbildung 6 zeigt die Ergebnistabelle.

	A	B	C	D	E	F	G
1		Objekt 1	Objekt 2	Objekt 3	Objekt 4	Objekt 5	fortlaufende Summe
2	Januar	335	333	350	326	357	1.701
3	Februar	282	365	396	332	238	3.314
4	März	343	219	214	380	308	4.778
5	April	294	218	304	229	373	6.196
6	Mai	309	281	333	361	277	7.757
7	Juni	377	282	210	237	285	9.148

Abb. 6: Die Ergebnistabelle mit der fortlaufenden Summierung

Mischbezüge

Neben den reinen relativen und absoluten Bezügen gibt es auch die *Mischbezüge*. Dabei ist ein Teil des Bezugs relativ, der andere absolut. Bei der absoluten Angabe wird vor dem Spaltenbuchstaben oder vor der Zeilennummer das Dollarzeichen geschrieben. Bei der jeweils anderen Angabe fehlt das Dollarzeichen. Auch hier soll der Einsatz eines Mischbezugs an einem konkreten Beispiel gezeigt werden. Dazu wird eine zweidimensionale Tabelle benötigt. Wobei in diesem einfachen Fall einfach nur das kleine Einmaleins von Excel berechnet werden soll (mathematisch ausgedrückt: $f_{xy} = x * y$). Die x-Werte (z.B. **1** bis **10**) werden in der Tabelle nebeneinander in einer Zeile eingetragen und die y-Werte (ebenfalls **1** bis **10**) untereinander in einer Spalte. Den Aufbau der Tabelle sehen Sie in Abbildung 7, Seite 7.

Nun sollen in die Tabellenzellen **C3** bis **L12** die Ergebnisse der Berechnung eingetragen werden. Zunächst wählen Sie die Tabellenzelle **C3** aus und tragen die Formel

=C\$2*\$B3

ein und bestätigen die Eingabe. Anschließend kopieren Sie die Formel mit dem *automatischen Ausfüllen* (siehe Skript **Excel für Microsoft 365 – Automatisches Ausfüllen**) in die Tabellenzellen **C4** bis **C12**. Lassen Sie den Zellbereich **C3** bis **C12** markiert und kopieren die Formel mit dem *automatischen Ausfüllen* nach rechts bis zur Spalte **L**. Das Ergebnis sehen Sie in Abbildung 8, Seite 7.

	A	B	C	D	E	F	G	H	I	J	K	L
1			x-Werte									
2			1	2	3	4	5	6	7	8	9	10
3	y-Werte	1										
4		2										
5		3										
6		4										
7		5										
8		6										
9		7										
10		8										
11		9										
12		10										

Abb. 7: Die Ausgangstabelle für das kleine Einmaleins

	A	B	C	D	E	F	G	H	I	J	K	L
1			x-Werte									
2			1	2	3	4	5	6	7	8	9	10
3	y-Werte	1	1	2	3	4	5	6	7	8	9	10
4		2	2	4	6	8	10	12	14	16	18	20
5		3	3	6	9	12	15	18	21	24	27	30
6		4	4	8	12	16	20	24	28	32	36	40
7		5	5	10	15	20	25	30	35	40	45	50
8		6	6	12	18	24	30	36	42	48	54	60
9		7	7	14	21	28	35	42	49	56	63	70
10		8	8	16	24	32	40	48	56	64	72	80
11		9	9	18	27	36	45	54	63	72	81	90
12		10	10	20	30	40	50	60	70	80	90	100

Abb. 8: Die komplette Tabelle mit dem kleinen Einmaleins

Bei der ersten Angabe der Formel steht vor der Zeilennummer das Dollarzeichen, aber nicht vor dem Spaltenbuchstaben. Das bedeutet, die Spaltenangabe ist relativ und wird bei Bedarf automatisch geändert, während die Zeilennummer absolut ist und sich nicht ändert. Warum diese Angabe so gemacht worden ist, können Sie leicht an den verschiedenen x-Werten in der Tabelle erkennen. Die Werte stehen in den Tabellenzellen **C2**, **D2**, **E2**, **F2**, usw. bis **L2**. Es sind also verschiedene Spaltenbuchstaben (**C**, **D**, **E**, **F**, usw.), aber immer dieselbe Zeilennummer (**2**). Beim späteren Kopieren der Formeln muss sich der Spaltenbuchstabe ändern (→ relativ), während die Zeilennummer sich nicht ändert (→ absolut).

Umgekehrt ist es bei den y-Werten. Diese stehen in den Tabellenzellen **B3**, **B4**, **B5**, **B6**, usw. bis **B12**. In diesem Fall ist es immer derselbe Spaltenbuchstabe (**B**), aber unterschiedliche Zeilennummern (**3**, **4**, **5**, **6**, usw.). Daher steht bei der zweiten Angabe in der Formel das Dollarzeichen vor dem Spaltenbuchstaben, aber nicht vor der Zeilennummer.

Anmerkung: Sie können natürlich auch komplexere Berechnungen als nur $x*y$ durchführen (z.B. $f_{xy} = \frac{\sin x}{\cos y}$ oder $f_{xy} = \frac{x^2}{1+y^3}$). Bezogen auf die Tabelle aus Abbildung 7 würde die Formel in der Tabellenzelle **C3** für das erste Beispiel **=SIN(C\$2)/COS(\$B3)** lauten. Für das zweite Beispiel lautet die Formel **=C\$2^2/(1+\$B3^3)**.

Schnittmengen berechnen

Mit Hilfe der relativen Bezüge stellt Excel auch die Möglichkeit dar, Werte aus einer Schnittmenge für eine Berechnung zu verwenden. Als Beispiel soll die Tabelle aus Abbildung 5, Seite 5, genommen und die Summe¹ einer bestimmten Schnittmenge berechnet werden. Nun wird eine freie Tabellenzelle (z.B. **B9**) ausgewählt und folgende Formel eingetragen (**beachten Sie bitte, dass sich zwischen den beiden angegebenen Zellbereichen C2:E4 und D3:F5 ein Leerzeichen befindet**):

=SUMME(C2:E4 D3:F5)

Nach Bestätigung der Eingabe erhalten Sie als Ergebnis den Wert **1.199**. Dieser bildet sich aus den Werten der Tabellenzellen **D3**, **D4**, **E3** und **E4**. Diese vier Tabellenzellen sind sowohl im Zellbereich **C2:E4**, als auch im Zellbereich **D3:F5** enthalten. Sie bilden also die Schnittmenge aus beiden Zellbereichen. Abbildung 9 zeigt nochmals grafisch die Schnittmenge aus beiden Zellbereichen.

	A	B	C	D	E	F
1		Objekt 1	Objekt 2	Objekt 3	Objekt 4	Objekt 5
2	Januar	335	333	350	326	357
3	Februar	282	365	396	332	238
4	März	343	219	214	380	308
5	April	294	218	304	229	373
6	Mai	309	281	333	361	277
7	Juni	377	282	210	237	285

Abb. 9: Die Schnittmenge zweier Zellbereiche

Anmerkung: Anstelle der Formel **=SUMME(C2:E4 D3:F5)** können Sie auch **=SUMME(D:E 3:4)** als Formel verwenden. Zwar haben Sie damit zwei komplette Spalten und zwei komplette Zeilen angegeben, aber die einzigen Tabellenzellen, die sowohl in den Spalten **D** und **E**, als auch in den Zeilen **3** und **4** liegen, sind die Tabellenzellen **D3**, **D4**, **E3** und **E4**.

Anstelle der Funktion **SUMME** können Sie auch (fast) jede andere Excel-Funktion verwenden. Wenn Sie beispielsweise den Mittelwert der Schnittmenge berechnen wollen, lautet die Formel: **=MITTELWERT(C2:E4 D3:F5)**.

Die Z1S1-Bezugsart

Bei der Z1S1-Bezugsart handelt es sich nicht um eine besondere Form der relativen oder absoluten Bezugsart, sondern es ist lediglich eine andere Darstellungsform bei der Angabe von Zellennamen in Formeln. Standardmäßig wird der Zellename aus dem Spaltenbuchstaben und der Zeilennummer gebildet. Bei der Z1S1-Bezugsart wird zunächst die Zeilennummer mit einem vorangestellten **Z** und dann die Spaltennummer mit einem vorangestellten **S** dargestellt. Die Tabellenzelle **F7** lautet dann in der Z1S1-Bezugsart: **Z7S6**. Mit Hilfe der Z1S1-Bezugsart lassen sich relative und absolute Bezüge noch deutlicher darstellen. Um allerdings die Z1S1-Bezugsart nutzen zu können, müssen Sie zunächst ein Kontrollkästchen bei den Excel-Optionen aktivieren. Wählen Sie im Register **Datei** den Befehl **Optionen**. Im Dialogfeld **Excel-Optionen** wählen Sie die Kategorie **For-**

Datei

¹ Es muss natürlich nicht immer die Summe von irgendwelchen Werten sein. Sie können auch andere Berechnungen (z.B. Mittelwert, Maximum, Minimum, usw.) durchführen.

meln und aktivieren in der Gruppe **Arbeiten mit Formeln** das Kontrollkästchen **Z1S1-Bezugsart** (siehe Abbildung 10) und bestätigen das Dialogfeld. Dass die Z1S1-Bezugsart aktiviert worden ist, können Sie an der Spaltenbeschriftung erkennen. Anstelle der üblichen Buchstaben sehen Sie Zahlen, genauso wie bei den Zeilennummern.





Abb. 10: Dialogfeld **Excel-Optionen**, Kategorie **Formeln**

Wie hilft nun diese spezielle Bezugsart bei der Unterscheidung von relativen und absoluten Bezügen? Schauen wir uns dazu noch mal die Tabelle mit den berechneten Arbeitszeiten und Arbeitslöhnen (siehe Abbildung 4, Seite 5) an. Wenn Sie bei einer Formel in die Zellenbearbeitung wechseln (z.B. mit der Funktionstaste ) , sehen Sie jetzt die Zellennamen in der Z1S1-Bezugsart (siehe Abbildung 11).

	1	2	3	4	5	6	7	8	9
1	Stundenlohn:	14,50 €		Mittagspause:	45 Min.		Inhalt von E1: 00:45:00		
2							Zahlenformat von E1: mm "Min."		
3	Mitarbeiter*in	Arbeitsbeginn	Arbeitsende	reine Arbeitszeit	Arbeitslohn				
4	Schmidt	07:45	15:55	=ZS(-1)-ZS(-2)-Z1S5					
5	Becker	07:41	15:40	07:14	104,88 €				
6	Mayer	08:01	16:10	07:24	107,30 €				
7	Volkman	08:05	16:05	07:15	105,13 €				
8	Müller	07:37	15:32	07:10	103,92 €				

Abb. 11: Eine Formel in der Z1S1-Bezugsart

Bei der ersten Zellangabe (das ist die Tabellenzelle **C4**) sehen Sie **ZS(-1)** in der Formel. Das bedeutet, es wird Bezug auf eine Tabellenzelle genommen, die sich in derselben Zeile befindet (hinter dem Buchstaben **Z** wird keine Zahl angezeigt) und eine Spalte links davon (das ist die Angabe **(-1)** hinter dem Buchstaben **S**). Analog dazu die zweite Angabe in der Formel (entspricht **B4**). Auch hier wieder dieselbe Zeile (nur der Buchstabe **Z**) und zwei Spalten links davon (**(-2)** hinter dem Buchstaben **S**). Bei der dritten Angabe handelt es sich in der „normalen“ Bezugsart um den absoluten Bezug **\$E\$1**. In der Z1S1-Bezugsart sehen Sie jetzt die Angabe **Z1S5**. Dadurch, dass die Zahlenangaben (hinter den Buchstaben **Z** bzw. **S**) nicht in runde Klammern eingeschlossen sind, können Sie erkennen, dass es sich um einen absoluten Bezug handelt.

Relative Bezüge in Formeln können Sie also daran erkennen, dass entweder hinter dem Buchstaben **Z** oder hinter dem Buchstaben **S** keine Zahl zu sehen ist (wenn die Formel in derselben Zeile oder Spalte steht, wo sich auch die Tabellenzelle befindet, auf die Bezug genommen wird), oder es werden positive Zahlen oder negative Zahlen (mit Minuszeichen) in runden Klammern eingeschlossen angezeigt. Steht kein Zeichen vor der Zahl (es handelt sich um eine positive Zahl), liegt die Tabellenzelle rechts oder unterhalb der Tabellenzelle mit der Formel. Bei einem Minuszeichen vor der Zahl liegt die Tabellenzelle links oder oberhalb der Tabellenzelle mit der Formel. Die Zahl selbst gibt die genaue Position an. Wenn beispielsweise **Z(3)S(-4)** angezeigt wird, bedeutet das (ausgehend von der Tabellenzelle mit der Formel) drei Zeilen weiter unten und vier Spalten weiter links. Angenommen, die Formel steht in der Tabellenzelle **F5** (bzw. **Z5S6**), dann bezieht sich die Angabe **Z(3)S(-4)** auf die Tabellenzelle **B8** (bzw. **Z8S2**).

In welcher Darstellung (normal oder Z1S1-Bezugsart) Sie in Excel arbeiten und Formeln angeben, spielt keine Rolle. Es ist allerdings etwas gewöhnungsbedürftig, in der Z1S1-Bezugsart zu arbeiten. Wenn Sie in die „normale“ Darstellung zurückwollen, müssen Sie im Dialogfeld **Excel-Optionen** nur das Kontrollkästchen **Z1S1-Bezugsart** deaktivieren.